
ReadMe File to accompany the dataset and Stata code for the paper “The Effect of Corporate Taxes on 


Investment and Entrepreneurship”: 


 


Unzip both the “Data.dta” and the “Corporate_Taxes.do” files into the directory you wish to use. 


Edit the Corporate_Taxes.do file to reflect the directory in which you placed the two files: 


• Replace the placeholder DIRECTORY in the fourth line of the Corporate_Taxes.do file with the 


path of the directory in which you have placed the two files. 


The Corporate_Taxes.do file contains all the code necessary to obtain the results for: 


• Tables 3 through 7 of the paper. 


• All the figures (Figures 1 through 6). 


• All the tables contained in the Web Appendices B and C. 


Please note that the Corporate_Taxes.do file makes use of variable lists both as local and global macros 


in order to avoid excessive repetition of code. The file does not contain any code to capture the Stata 


output in a form other than the log file.   







Corporate_Taxes.do
capture log close
clear
set mem 500M
cd "DIRECTORY"
log using Corporate_Taxes.txt, replace text
use Data

set more off


***Definition of global variablelists***
global taxvars statutory effective effective_5yr
global outcomes Investment2005 FDI2005 nbusinessdensitycf nentryrateav


***************************************
*               TABLE 3                *
***************************************

local income UpperMiddleIncome LowerMiddleIncome LowIncome

***Corporate tax rates***
bysort incomegroup: sum $taxvars
sum $taxvars

foreach item in `income'{
	foreach varname of varlist $taxvars{
		ttest `varname' if (incomegroup=="HighIncome" | incomegroup=="`item'"), by(incomegroup)
		}
	}

***Other tax rates***
local othertaxrates labor other_taxes vatsales pit2004
bysort incomegroup: sum `othertaxrates'
sum `othertaxrates'

foreach item in `income'{
	foreach varname of varlist `othertaxrates'{
		ttest `varname' if (incomegroup=="HighIncome" | incomegroup=="`item'"), by(incomegroup)
		}
	}

***Tax administration***
local taxadmin payments2004 time2004
bysort incomegroup: sum `taxadmin'
sum `taxadmin'

foreach item in `income'{
	foreach varname of varlist `taxadmin'{
		ttest `varname' if (incomegroup=="HighIncome" | incomegroup=="`item'"), by(incomegroup)
		}
	}

***Investment and entrepreneurship***
local inv_ent Investment2005 FDI2005 nbusinessdensitycf nentryrateav
bysort incomegroup: sum `inv_ent'
sum `inv_ent'

foreach item in `income'{
	foreach varname of varlist `inv_ent'{
		ttest `varname' if (incomegroup=="HighIncome" | incomegroup=="`item'"), by(incomegroup)
		}
	}

***Other dependent variables***
local otherdep avinformal3 dew2004
bysort incomegroup: sum `otherdep' MEInv_manuf MEInv_servi
sum `otherdep' MEInv_servi MEInv_manuf

foreach item in `income'{
	foreach varname of varlist `otherdep'{
		ttest `varname' if (incomegroup=="HighIncome" | incomegroup=="`item'"), by(incomegroup)
		}
	}


***************************************
*               TABLE 4               *
***************************************

*Note that we keep using the local variable lists defined above for table 3.

***Corporate tax rates***
bysort dblegor: sum $taxvars
sum $taxvars
foreach varname of varlist $taxvars{
	ttest `varname' if (dblegor=="English" | dblegor=="French"), by(dblegor)
	}

***Other tax rates***
bysort dblegor: sum `othertaxrates'
sum `othertaxrates'
foreach varname of varlist `othertaxrates'{
	ttest `varname' if (dblegor=="English" | dblegor=="French"), by(dblegor)
	}

***Tax administration***
bysort dblegor: sum `taxadmin'
sum `taxadmin'
foreach varname of varlist `taxadmin'{
	ttest `varname' if (dblegor=="English" | dblegor=="French"), by(dblegor)
	}

***Investment and entrepreneurship***
bysort dblegor: sum `inv_ent'
sum `inv_ent'
foreach varname of varlist `inv_ent'{
	ttest `varname' if (dblegor=="English" | dblegor=="French"), by(dblegor)
	}

***Other dependent variables***
local otherdep avinformal3 dew2004 MEInv_manuf MEInv_servi
bysort dblegor: sum `otherdep' 
sum `otherdep' 
foreach varname of varlist `otherdep'{
	ttest `varname' if (dblegor=="English" | dblegor=="French"), by(dblegor)
	}


***************************************
*               TABLE 5               *
***************************************
foreach varname1 of varlist $outcomes{
	foreach varname2 of varlist $taxvars{
		reg `varname1' `varname2'
		}
	}


***************************************
*               TABLE 5a              *
***************************************
local controls5a other_taxes vatsales pit2004

foreach varname1 of varlist $outcomes{
	foreach varname2 of varlist $taxvars{
		reg `varname1' `varname2' `controls5a'
		}
	}


***************************************
*               TABLE 5b              *
***************************************
local controls5b lnpayments2004 index_gcr sb_proc2004

foreach varname1 of varlist $outcomes{
	foreach varname2 of varlist $taxvars{
		reg `varname1' `varname2' `controls5b'
		}
	}


***************************************
*               TABLE 5c              *
***************************************
local controls5c propertyrights emplo_i2004 freedomtotradeinternationally lngdppc2003

foreach varname1 of varlist $outcomes{
	foreach varname2 of varlist $taxvars{
		reg `varname1' `varname2' `controls5c'
		}
	}


***************************************
*               TABLE 5d              *
***************************************
#delimit ;
local controls5d 	other_taxes vatsales pit2004 lnpayments2004 lngdppc2003 propertyrights sb_proc2004 
			emplo_i2004 freedomtotradeinternationally seign2004 inf10yearavg index_gcr;
#delimit cr

foreach varname1 of varlist $outcomes{
	foreach varname2 of varlist $taxvars{
		reg `varname1' `varname2' `controls5d'
		}
	}


***************************************
*               TABLE 6               *
***************************************

***PanelA***
foreach varname2 of varlist $taxvars{
	reg MEInv_manuf `varname2'
	}
foreach varname2 of varlist $taxvars{
	reg MEInv_servi `varname2'
	}

***PanelB***
foreach varname2 of varlist $taxvars{
	reg avinformal3 `varname2' lngdppc2003
	}
foreach varname2 of varlist $taxvars{
	reg avinformal3 `varname2' index_gcr
	}


***************************************
*               TABLE 7               *
***************************************
foreach varname2 of varlist $taxvars{
	reg dew2004 `varname2' EqSec_GDP_2003 lngdppc2003 if country!="Zimbabwe"
	}


***************************************
*               FIGURES               *
***************************************

***Figure1***
twoway (lfit Investment2005 effective) (scatter Investment2005 effective, mlabel(code)), title(1st Year Effective Tax Rate and Investment) xtitle(1st Year Effective Tax Rate ) ytitle(Investment 2003-2005 as % of GDP) scheme(s2color) legend(off) 

***Figure2***
twoway (lfit FDI2005 effective) (scatter FDI2005 effective, mlabel(code)), title(1st Year Effective Tax Rate and FDI) xtitle(1st Year Effective Tax Rate ) ytitle(FDI 2003-2005 as % of GDP) scheme(s2color) legend(off)

***Figure3***
twoway (lfit nbusinessdensitycf effective) (scatter nbusinessdensitycf effective, mlabel(code)), title(1st Year Effective Tax Rate and Business Density) xtitle(1st Year Effective Tax Rate ) ytitle(Business Density per 100 people) scheme(s2color) legend(off)

***Figure4***
twoway (lfit nentryrateav effective) (scatter nentryrateav effective, mlabel(code)), title(1st year effective tax rate and Entry rate) xtitle(1st Year Effective Tax Rate ) ytitle(Average Entry Rate (%) 2000-2004) scheme(s2color) legend(off)

***Figure5***
twoway (lfit avinformal3 effective) (scatter avinformal3 effective, mlabel(code)), title(1st year effective tax rate and Size of informal sector) xtitle(1st Year Effective Tax Rate ) ytitle(Size of the Informal Sector 2005-07) scheme(s2color) legend(off)

***Figure6***
twoway (lfit dew2004 effective) (scatter dew2004 effective, mlabel(code)), title(1st year effective tax rate and Debt to equity ratio) xtitle(1st Year Effective Tax Rate ) ytitle(Debt-to-Equity Ratio) scheme(s2color) legend(off)


***************************************
*          WEB APPENDIX TABLES        *
***************************************

***Web Appendix B, restricted to Sample 1
foreach varname1 of varlist $outcomes{
	foreach varname2 of varlist $taxvars{
		reg `varname1' `varname2' if legalform==0
		}
	}

***Web Appendix B, restricted to Sample 2
foreach varname1 of varlist $outcomes{
	foreach varname2 of varlist $taxvars{
		reg `varname1' `varname2' if legalform==1
		}
	}

***Web Appendix B, restricted to Sample 3
foreach varname1 of varlist $outcomes{
	foreach varname2 of varlist $taxvars{
		reg `varname1' `varname2' if legalform==.
		}
	}

***Web Appendix C
foreach varname2 of varlist $taxvars{
	reg FDIoecd2004 `varname2'
	}


clear
log close



